
AVR108 : LPM命令の初期設定と使用

序説
この応用記述はAVRﾏｲｸﾛ ｺﾝﾄﾛｰﾗのﾌﾟﾛｸﾞﾗﾑ用ﾌﾗｯｼｭ ﾒﾓﾘに格納された定数の入出力方法
を記述します。AVRはﾊｰﾊﾞｰﾄﾞ構造を基にし、これはﾒﾓﾘのｱﾄﾞﾚｽとﾃﾞｰﾀが独立したﾊﾞｽを使用
することを意味します。これは単一周期命令実行速度を成し遂げるために必要です。ﾌﾗｯｼｭ ﾒ
ﾓﾘに定数を格納できるよう、命令ｾｯﾄにLPM(Load Program Memory)命令が含まれています。

LPMの使用
ﾌﾟﾛｸﾞﾗﾑ用ﾌﾗｯｼｭ ﾒﾓﾘからﾚｼﾞｽﾀ ﾌｧｲﾙ内にﾊﾞｲﾄ ﾃﾞｰﾀを取得するため、AVR命令ｾｯﾄにLPM
命令が含まれます。

AVRﾏｲｸﾛ ｺﾝﾄﾛｰﾗのﾌﾟﾛｸﾞﾗﾑ用ﾌﾗｯｼｭ ﾒﾓﾘは16ﾋﾞｯﾄ語として構成されます。ﾚｼﾞｽﾀ ﾌｧｲﾙと
ﾃﾞｰﾀ用SRAMは8ﾋﾞｯﾄのﾊﾞｲﾄとして構成されます。従ってﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘからﾃﾞｰﾀ用ﾒﾓﾘにﾃﾞｰﾀ
を取得する時に特別な考慮が祓われなければなりません。

ﾚｼﾞｽﾀ ﾌｧｲﾙのZﾚｼﾞｽﾀがﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘをｱｸｾｽするのに使用されます。この16ﾋﾞｯﾄ ﾚｼﾞｽﾀ対
はﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘへの16ﾋﾞｯﾄ ﾎﾟｲﾝﾀとして使用されます。上位15ﾋﾞｯﾄはﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘの語ｱﾄﾞﾚ
ｽを選択します。このため、語ｱﾄﾞﾚｽはそれがZﾚｼﾞｽﾀに置かれる前に2倍されます。

要点
■ AVR®ｱｾﾝﾌﾞﾗとでのLPM(Load Program Memory)命令の使用
■ ﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘからの定数取得
■ 参照表の使用

ﾌﾟﾛｸﾞﾗﾑ用ﾌﾗｯｼｭ ﾒﾓﾘ 語ｱﾄﾞﾚｽ

図1. Z ｱﾄﾞﾚｽ ﾚｼﾞｽﾀ

上位/下位
ﾊﾞｲﾄ選択

Zｱﾄﾞﾚｽ ﾚｼﾞｽﾀの最下位ﾋﾞｯﾄ(LSB)はﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘ語の上位(1)または下位(0)のどちらかを選
択します。ｱﾄﾞﾚｽの上位(ZH)と下位(ZL)を計算するには、HIGH()関数とLOW()関数を使用して
ください。

ﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘの任意順の位置からﾃﾞｰﾀを取得するには、新しいｱﾄﾞﾚｽがｱｸｾｽされる度毎にZ
ﾚｼﾞｽﾀが正しいｱﾄﾞﾚｽで設定されなければなりません。

ﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘではﾃﾞｰﾀがﾌﾟﾛｸﾞﾗﾑ語内で下位部の1ﾊﾞｲﾄと上位部の次ﾊﾞｲﾄとで構成されてい
ます。このため、ﾒｯｾｰｼﾞ文字列はAVR Studio®内のﾒﾓﾘ ｳｨﾝﾄﾞｳで見る時に全ての文字の対が
入れ替わった場合のように現れます。

この応用記述内のﾌﾟﾛｸﾞﾗﾑはﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘから文字列のﾊﾞｲﾄを取得し、それをﾎﾟｰﾄBに書きま
す。最初に全ﾋﾟﾝが出力となるようにﾎﾟｰﾄBを初期化します。文字列"Hello World"の開始ｱﾄﾞﾚｽ
を上記のようにZﾚｼﾞｽﾀに設定します。そしてLPM命令を使用してﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘからﾊﾞｲﾄが取
得されます。ﾌﾟﾛｸﾞﾗﾑは文字列の最後に達したか(ﾊﾞｲﾄが0だったか)そうでないかどうかを調べ
ます。未だ最後に達していない場合、最後に読んだﾊﾞｲﾄがﾎﾟｰﾄBに置かれ、短い遅延が作ら
れ、そしてZﾚｼﾞｽﾀが進行(+1)されます。その後、ﾌﾟﾛｸﾞﾗﾑは別のﾊﾞｲﾄを取得するために元の方
向へ飛びます。

;****  A V R 1 0 8 応用記述  *********************************************************
;*
;* 表題: Load Program Memory
;* 版: 1.0
;* 最終更新: 98.12.17
;* 対象: AT90Sxx1xとより上位機種 (SRAM付きﾃﾞﾊﾞｲｽ)
;*
;* 支援E-mail: avr@atmel.com
;*
;* 内容
;* この応用記述はLPM(Load Program Memory)命令の使用法を示します。
;* 応用記述はﾌﾟﾛｸﾞﾗﾑ用ﾒﾓﾘから文字列"Hello World"をﾊﾞｲﾄ単位で取得し、それをﾎﾟｰﾄBに置きます。
;*
;**************************************************************************************

 .INCLUDE "8515def.inc" ; ﾃﾞﾊﾞｲｽ定義ｲﾝｸﾙｰﾄﾞ ﾌｧｲﾙ指定
 .DEVICE AT90S8515 ; ﾃﾞﾊﾞｲｽ指定　(注: 現ｱｾﾝﾌﾞﾗでは無効)
 temp =R16 ; 一時変数定義

start: LDI temp,HIGH(RAMEND) ; 内部SRAM最終位置上位値取得
 OUT SPH,temp ; ｽﾀｯｸ ﾎﾟｲﾝﾀ上位初期化
 LDI temp,LOW(RAMEND) ; 内部SRAM最終位置下位値取得
 OUT SPL,temp ; ｽﾀｯｸ ﾎﾟｲﾝﾀ上位初期化
;
 LDI temp,0b11111111 ; 全1値取得
 OUT PORTB,temp ; ﾎﾟｰﾄB全ﾋﾟﾝHigh設定
 OUT DDRB,temp ; ﾎﾟｰﾄB全ﾋﾟﾝ出力設定

; Zﾚｼﾞｽﾀ内に'msg'のｱﾄﾞﾚｽを取得(ﾊﾞｲﾄ ｱﾄﾞﾚｽにするために語ｱﾄﾞﾚｽを2倍)
; 上位/下位ｱﾄﾞﾚｽ ﾊﾞｲﾄ計算にHIGH/LOW関数を使用

 LDI ZH,HIGH(msg*2) ; 文字列先頭ﾊﾞｲﾄ ｱﾄﾞﾚｽ上位値設定
 LDI ZL,LOW(msg*2) ; 文字列先頭ﾊﾞｲﾄ ｱﾄﾞﾚｽ下位値設定
llp: LPM  ; 文字列から対応ﾊﾞｲﾄをR0に取得
 TST R0 ; 文字列終了子(=0)検査
 BREQ quit ; 終了子なら無限繰り返しへ分岐
;
 OUT PORTB,R0 ; 対応ﾊﾞｲﾄ値をﾎﾟｰﾄBに出力
 RCALL delay ; 約1s待機
 ADIW ZH:ZL,1 ; 文字列ﾎﾟｲﾝﾀ進行(+1)
 RJMP llp ; 終了子まで継続へ
;
quit: RJMP quit ; 無限繰り返し(待機)

; ( 約1s待機(4MHzｼｽﾃﾑ ｸﾛｯｸ時) )

delay: LDI R20,20 ; 第3計数器初期値設定
 CLR R21 ; 第2計数器初期値設定
 CLR R22 ; 第1計数器初期値設定
dlp: DEC R22 ; 第1計数器減数(-1)
 BRNE dlp ; 約192μs経過まで繰り返し待機
;
 DEC R21 ; 第2計数器減数(-1)
 BRNE dlp ; 約49.2ms経過まで繰り返し待機
;
 DEC R20 ; 第3計数器減数(-1)
 BRNE dlp ; 約983ms経過まで繰り返し待機
;
 RET  ; 呼び出し元へ復帰

; * 定数 *

msg: .DB "Hello World" ; 文字列定数定義
 .DB 0 ; 文字列終了子定義

本応用記述はATMELのAVR108応用記述(doc1233.pdf Rev.1233B-05/02)の翻訳日本語版です。日本語では不自然となる重複す
る形容表現は省略されている場合があります。日本語では難解となる表現は大幅に意訳されている部分もあります。必要に応じて一
部加筆されています。頁割の変更により、原本より頁数が少なくなっています。

必要と思われる部分には( )内に英語表記や略称などを残す形で表記しています。

青字の部分はﾘﾝｸとなっています。一般的に赤字の0,1は論理0,1を表します。その他の赤字は重要な部分を表します。

